

D01GBF – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

1 Purpose

D01GBF returns an approximation to the integral of a function over a hyper-rectangular region, using a Monte Carlo method. An approximate relative error estimate is also returned. This routine is suitable for low accuracy work.

2 Specification

```

SUBROUTINE D01GBF(NDIM, A, B, MINCLS, MAXCLS, FUNCTN, EPS, ACC,
1          LENWRK, WRKSTR, FINEST, IFAIL)
  INTEGER      NDIM, MINCLS, MAXCLS, LENWRK, IFAIL
  real        A(NDIM), B(NDIM), FUNCTN, EPS, ACC,
1          WRKSTR(LENWRK), FINEST
  EXTERNAL    FUNCTN

```

3 Description

D01GBF uses an adaptive Monte Carlo method based on the algorithm described by Lautrup [1]. It is implemented for integrals of the form:

$$\int_{a_1}^{b_1} \int_{a_2}^{b_2} \dots \int_{a_n}^{b_n} f(x_1, x_2, \dots, x_n) dx_n \dots dx_2 dx_1.$$

Upon entry, unless LENWRK has been set to the minimum value $10 \times \text{NDIM}$, the routine subdivides the integration region into a number of equal volume subregions. Inside each subregion the integral and the variance are estimated by means of pseudo-random sampling. All contributions are added together to produce an estimate for the whole integral and total variance. The variance along each co-ordinate axis is determined and the routine uses this information to increase the density and change the widths of the sub-intervals along each axis, so as to reduce the total variance. The total number of subregions is then increased by a factor of two and the program recycles for another iteration. The program stops when a desired accuracy has been reached or too many integral evaluations are needed for the next cycle.

4 References

- [1] Lautrup B (1971) An adaptive multi-dimensional integration procedure *Proc. 2nd Coll. Advanced Methods in Theoretical Physics, Marseille*

5 Parameters

- | | | |
|----|--|--------------|
| 1: | NDIM — INTEGER | <i>Input</i> |
| | <i>On entry:</i> the number of dimensions of the integral, n . | |
| | <i>Constraint:</i> $\text{NDIM} \geq 1$. | |
| 2: | A(NDIM) — <i>real</i> array | <i>Input</i> |
| | <i>On entry:</i> the lower limits of integration, a_i , for $i = 1, 2, \dots, n$. | |
| 3: | B(NDIM) — <i>real</i> array | <i>Input</i> |
| | <i>On entry:</i> the upper limits of integration, b_i , for $i = 1, 2, \dots, n$. | |

4: MINCLS — INTEGER *Input/Output*

On entry: MINCLS must be set:

either to the minimum number of integrand evaluations to be allowed, in which case $\text{MINCLS} \geq 0$;

or to a negative value. In this case the routine assumes that a previous call had been made with the same parameters NDIM, A and B and with either the same integrand (in which case D01GBF continues calculation) or a similar integrand (in which case D01GBF begins the calculation with the subdivision used in the last iteration of the previous call). See also WRKSTR.

On exit: MINCLS contains the number of integrand evaluations actually used by D01GBF.

5: MAXCLS — INTEGER *Input*

On entry: the maximum number of integrand evaluations to be allowed. In the continuation case this is the number of new integrand evaluations to be allowed. These counts do not include zero integrand values.

Constraints:

$$\begin{aligned} \text{MAXCLS} &> \text{MINCLS}, \\ \text{MAXCLS} &\geq 4 \times (\text{NDIM} + 1). \end{aligned}$$

6: FUNCTN — *real* FUNCTION, supplied by the user. *External Procedure*

FUNCTN must return the value of the integrand f at a given point.

Its specification is:

<i>real</i> FUNCTION FUNCTN(NDIM, X)
INTEGER NDIM
<i>real</i> X(NDIM)

1: NDIM — INTEGER *Input*

On entry: the number of dimensions of the integral, n .

2: X(NDIM) — *real* array *Input*

On entry: the co-ordinates of the point at which the integrand must be evaluated.

FUNCTN must be declared as EXTERNAL in the (sub)program from which D01GBF is called. Parameters denoted as *Input* must **not** be changed by this procedure.

7: EPS — *real* *Input*

On entry: the relative accuracy required.

Constraint: $\text{EPS} \geq 0.0$.

8: ACC — *real* *Output*

On exit: the estimated relative accuracy of FINEST.

9: LENWRK — INTEGER *Input*

On entry: the dimension of the array WRKSTR as declared in the (sub)program from which D01GBF is called..

For maximum efficiency, LENWRK should be about

$$3 \times \text{NDIM} \times (\text{MAXCLS}/4)^{1/\text{NDIM}} + 7 \times \text{NDIM}.$$

If LENWRK is given the value $10 \times \text{NDIM}$ then the subroutine uses only one iteration of a crude Monte Carlo method with MAXCLS sample points.

Constraint: $\text{LENWRK} \geq 10 \times \text{NDIM}$.

- 10:** WRKSTR(LENWRK) — *real* array *Input/Output*
On entry: if MINCLS < 0, WRKSTR must be unchanged from the previous call of D01GBF – except that for a new integrand WRKSTR(LENWRK) must be set to 0.0. See also MINCLS.
On exit: WRKSTR contains information about the current sub-interval structure which could be used in later calls of D01GBF. In particular, WRKSTR(*j*) gives the number of sub-intervals used along the *j*th co-ordinate axis.
- 11:** FINEST — *real* *Output*
On exit: the best estimate obtained for the integral.
- 12:** IFAIL — INTEGER *Input/Output*
On entry: IFAIL must be set to 0, -1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.
On exit: IFAIL = 0 unless the routine detects an error or gives a warning (see Section 6).
For this routine, because the values of output parameters may be useful even if IFAIL ≠ 0 on exit, users are recommended to set IFAIL to -1 before entry. **It is then essential to test the value of IFAIL on exit.** To suppress the output of an error message when soft failure occurs, set IFAIL to 1.

6 Error Indicators and Warnings

Errors or warnings specified by the routine:

IFAIL = 1

- On entry, NDIM < 1,
- or MINCLS ≥ MAXCLS,
- or LENWRK < 10 × NDIM,
- or MAXCLS < 4 × (NDIM+1),
- or EPS < 0.0.

IFAIL = 2

MAXCLS was too small for D01GBF to obtain the required relative accuracy EPS. In this case D01GBF returns a value of FINEST with estimated relative error ACC, but ACC will be greater than EPS. This error exit may be taken before MAXCLS non-zero integrand evaluations have actually occurred, if the routine calculates that the current estimates could not be improved before MAXCLS was exceeded.

7 Accuracy

A relative error estimate is output through the parameter ACC. The confidence factor is set so that the actual error should be less than ACC 90% of the time. If a user desires a higher confidence level then a smaller value of EPS should be used.

8 Further Comments

The running time for D01GBF will usually be dominated by the time used to evaluate the integrand FUNCTN, so the maximum time that could be used is approximately proportional to MAXCLS.

For some integrands, particularly those that are poorly behaved in a small part of the integration region, D01GBF may terminate with a value of ACC which is significantly smaller than the actual relative error. This should be suspected if the returned value of MINCLS is small relative to the expected difficulty of the integral. Where this occurs, D01GBF should be called again, but with a higher entry value of MINCLS (e.g., twice the returned value) and the results compared with those from the previous call.

The exact values of FINEST and ACC on return will depend (within statistical limits) on the sequence of random numbers generated within D01GBF by calls to G05CAF. Separate runs will produce identical answers unless the part of the program executed prior to calling D01GBF also calls (directly or indirectly) routines from the G05 Chapter Introduction, and the series of such calls differs between runs. If desired, the user may ensure the identity or difference between runs of the results returned by D01GBF, by calling G05CBF or G05CCF respectively, immediately before calling D01GBF.

9 Example

This example program calculates the integral

$$\int_0^1 \int_0^1 \int_0^1 \int_0^1 \frac{4x_1 x_3 \exp(2x_1 x_3)}{(1 + x_2 + x_4)^2} dx_1 dx_2 dx_3 dx_4 = 0.575364.$$

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```

*      D01GBF Example Program Text
*      Mark 14 Revised.  NAG Copyright 1989.
*      .. Parameters ..
INTEGER          NDIM, MAXCLS, LENWRK
PARAMETER       (NDIM=4,MAXCLS=20000,LENWRK=500)
INTEGER          NOUT
PARAMETER       (NOUT=6)
*      .. Local Scalars ..
real           ACC, EPS, FINEST
INTEGER          IFAIL, K, MINCLS
*      .. Local Arrays ..
real           A(NDIM), B(NDIM), WRKSTR(LENWRK)
*      .. External Functions ..
real           FUNCTN
EXTERNAL         FUNCTN
*      .. External Subroutines ..
EXTERNAL         D01GBF
*      .. Executable Statements ..
WRITE (NOUT,*) 'D01GBF Example Program Results'
DO 20 K = 1, NDIM
    A(K) = 0.0e0
    B(K) = 1.0e0
20 CONTINUE
EPS = 0.01e0
MINCLS = 1000
IFAIL = 1
*
CALL D01GBF(NDIM,A,B,MINCLS,MAXCLS,FUNCTN,EPS,ACC,LENWRK,WRKSTR,
+          FINEST,IFAIL)
*
WRITE (NOUT,*)
IF (IFAIL.GT.0) THEN
    WRITE (NOUT,99999) 'D01GBF fails. IFAIL =', IFAIL
    WRITE (NOUT,*)
END IF
IF (IFAIL.EQ.0 .OR. IFAIL.EQ.2) THEN
    WRITE (NOUT,99998) 'Requested accuracy   = ', EPS
    WRITE (NOUT,99997) 'Estimated value     = ', FINEST
    WRITE (NOUT,99998) 'Estimated accuracy = ', ACC

```

```

        WRITE (NOUT,99999) 'Number of evaluations = ', MINCLS
    END IF
    STOP
*
99999 FORMAT (1X,A,I5)
99998 FORMAT (1X,A,e13.2)
99997 FORMAT (1X,A,F13.5)
    END
*
    real FUNCTION FUNCTN(NDIM,X)
*
    .. Scalar Arguments ..
    INTEGER          NDIM
*
    .. Array Arguments ..
    real           X(NDIM)
*
    .. Intrinsic Functions ..
    INTRINSIC       EXP
*
    .. Executable Statements ..
    FUNCTN = 4.0e0*X(1)*X(3)**2*EXP(2.0e0*X(1)*X(3))/(1.0e0+X(2)+X(4))
+          **2
    RETURN
    END

```

9.2 Program Data

None.

9.3 Program Results

D01GBF Example Program Results

```

Requested accuracy   =      0.10E-01
Estimated value     =      0.57554
Estimated accuracy  =      0.82E-02
Number of evaluations = 1728

```
